Maple Commands for the Surface

```
restart;
with(plots):
f:=cos(beta)*(cos(alpha)+2);
g:=sin(beta)*(cos(alpha)+2);
h:=sin(alpha);
plot3d([f,g,h],alpha=0..2*Pi,beta=0..2*Pi,scaling=constrained,
    style=patchnogrid,shading=ZHUE);
```

Maple Commands for a Path

These commands assume the first section has been done. We will find a parametric representation for the path- That means α, β are functions of time. Then these are substituted into the equations for the torus.

NOTE: The end of the line for each of these is for readability. Each of these would be typed on its own line.

For this example, $\alpha = 0$ and $\beta = \pi t$ so that, as t goes from 0 to 1, (α, β) goes from $(0, 0)$ to $(0, \pi)$. This path on the torus goes from $(3, 0, 0)$ to $(-3, 0, 0)$.

```
Path1Functions:=subs({alpha=0, beta=t*Pi},f), subs({alpha=0, beta=t*Pi},g),
        subs({alpha=0, beta=t*Pi},h);

Path1:=spacecurve([Path1Functions, t=0..1],color=black,thickness=5):

Torus1:=plot3d([f,g,h],alpha=0..2*Pi,beta=0..2*Pi,scaling=constrained,
        style=patchnogrid,shading=ZHUE):

display3d({Torus1, Path1});
```

The Arc Length

If we assume that α, β are functions of time, $0 \leq t \leq 1$, and these have been substituted into the equations for x, y and z for the surface of the torus, then we can think of x, y and z as functions of time. The arc length of this path in 3-d is:

$$\int_0^1 \sqrt{\left(\frac{dx}{dt}\right)^2 + \left(\frac{dy}{dt}\right)^2 + \left(\frac{dz}{dt}\right)^2}\, dt$$

so, continuing our commands in Maple, we could write:

```
dx:=diff(Path1Functions[1],t);
dy:=diff(Path1Functions[2],t);
```

```
dz:=diff(Path1Functions[3],t);

PathLength:=Int(sqrt(dx^2+dy^2+dz^2),t=0..1);
evalf(PathLength);
```